/*
Copyright (c) 2009, Young Digital Planet SA
All rights reserved.

* Redistribution and use in source and binary forms, with or without 
  modification, are permitted provided that the following conditions are met:
* Redistributions of source code must retain the above copyright notice, this 
  list of conditions and the following disclaimer.
* Redistributions in binary form must reproduce the above copyright notice, 
  this list of conditions and the following disclaimer in the documentation 
  and/or other materials provided with the distribution.
* Neither the name of the Young Digital Planet SA nor the names of its 
  contributors may be used to endorse or promote products derived from this 
  software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" 
AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE 
IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE 
ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE 
LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR 
CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF 
SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS 
INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN 
CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) 
ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE 
POSSIBILITY OF SUCH DAMAGE.
*/
package pl.ydp.nexl.impl
{
	import pl.ydp.nexl.INexlAttributes;

	public class NexlAttributes implements INexlAttributes
	{
		protected var m_atts:Object;
		
		public function NexlAttributes()
		{
		}
		
		public function get attributesNames():Object {
			return m_atts;
		}
		
		public function getAttributeValues( name:String ):* {
			var cVal:* = m_atts[name]; 
			if ( cVal is Att ) {
				return cVal.value;
			} else if ( cVal is Array ) {
				return cVal.map( function callback(item:Att, index:int, array:Array):* {
					return item.value;
				} );
			} else
				return null;
		}
		
		public function getAttributeString( name:String ):String {
			var val:* = getAttributeValues(name);
			
			return val==null ? 
					null :
					(val is Array ? 
						String(val[0]) : 
						String(val));	
		}
		
		public function getAttributeNumber( name:String ):Number {
			var val:* = getAttributeValues(name);
			
			return val==null ? 
					NaN :
					(val is Array ? 
						Number(val[0]) : 
						Number(val));	
		}
		
		public function attributeHasValue( name:String, value:String ):Boolean {
			var cVal:* = m_atts[name]; 
			if ( cVal is Att ) {
				return cVal.value==value;
			} else if ( cVal is Array ) {
				return cVal.some( function callback(item:Att, index:int, array:Array):Boolean {
					return item.value == value;
				} );
			} else
				return false;
		}
		
		internal function loadData( extNode:XML ):void {
			m_atts = { };
			
			for each( var att:XML in extNode.attr )
				appendAtt( att.@name, new Att(extractValue(att),att.@type) );
			var id:String = extNode.@id;
			if ( id!=null && id.length>0 )
				appendAtt( "id", new Att(id,"string") );
		}
		
		internal function get atts():Object {
			return m_atts;
		}
		
		protected function appendAtt( name:String, val:Att ):void {
			var cVal:* = m_atts[name]; 
			
			if ( cVal is Att ) {
				m_atts[name] = [cVal,val];
			} else if ( cVal is Array )
				m_atts[name].push( val );
			else
				m_atts[name] = val; 
		}
		
		internal function extractValue( node:XML ):String {
			if ( node==null )
				return null;
			else if ( node.hasOwnProperty("@value") )
				return String(node.@value);
			else {
				var ch:XMLList = node.children();
				if ( ch.length()==1 && ch[0].nodeKind()=="text" )
					return ch[0].toString();
				else
					return node.*.toXMLString();
			}
		}
	}
}

class Att {
	public var value:*;
	public var type:String;
	
	public function Att( val:*, type:String ) {
		value = val;
		type = type;
	}
}